
Nástroj CPack

• Modul systému CMake pro vytváření instalátorů a balíčků
• Podpora pro Windows a Linux (DEB a RPM)
• Pro vytvoření balíčků používá nástroje dané platformy

• Windows – NSIS
• Linux (RPM) – rpmbuild

• Integrován s CMake pomocí install cílů
• Balíček je možné vytvořit pomocí nového cíle: package

cmake_minimum_required(VERSION 3.0.0)
project(InstallerTest)

# install(...)

# Informace o balíčku (název, autor, ...)

# Nastavení specifického generátoru balíčku

include(CPack)

https://nsis.sourceforge.io/Main_Page


Informace o balíčku

• Základní informace o balíčku
• Společné pro všechny generátory

set(CPACK_PACKAGE_NAME "${PROJECT_NAME}")
set(CPACK_PACKAGE_VENDOR "ivs.fit.vutbr.cz")
set(CPACK_PACKAGE_CONTACT "iveigend@fit.vut.cz")
set(CPACK_PACKAGE_DESCRIPTION_SUMMARY "Installer Test")
set(CPACK_PACKAGE_VERSION "1.0.0")

• ${PROJECT_NAME} nastaven pomocí project()

Nástroj CPack – Windows

• Tvorba instalátoru pomocí CPack a NSIS
set(CPACK_GENERATOR NSIS) # Výběr generátoru NSIS

# Instalace všech souborů z "deployment"
# do hlavního adresáře instalace
install(DIRECTORY deployment/ DESTINATION .)

# Výchozí cesta instalace "Program Files/InstallerTest"
set(CPACK_PACKAGE_INSTALL_DIRECTORY "${PROJECT_NAME}")

# Přídavný NSIS skript spouštěný při instalaci
# (zde instaluje VC++ redistributable
# z hlavního adresáře instalace)
set(CPACK_NSIS_EXTRA_INSTALL_COMMANDS

"Execwait 'vcredist_x64.exe /q'")

Nástroj CPack – Windows

# Ruční vytvoření zástupce na pracovní ploše OS.
# Promněnné $DESKTOP a $INSTDIR jsou definovány
# až v době instalace NSISem
set(CPACK_NSIS_CREATE_ICONS_EXTRA

"CreateShortCut '$DESKTOP\\\\${PROJECT_NAME}.lnk'
'$INSTDIR\\\\${PROJECT_NAME}.exe'")

# Odstranění zástupců při odinstalaci
set(CPACK_NSIS_DELETE_ICONS_EXTRA

"Delete '$DESKTOP\\\\${PROJECT_NAME}.lnk'")

# \\\\ - několik parserů (CPack + NSIS)
# Příkazy "CreateShortCut" a "Delete"
# jsou součástí NSIS

• Generování balíčku (seznam generátorů: pouze parametr -G)

> cmake -G "Visual␣Studio␣17␣2022" .
> cmake --build . --target package

Nástroj CPack – Linux (DEB)

• Tvorba DEB balíčku pomocí CPack
set(CPACK_GENERATOR DEB) # Generátor DEB

# Opět chceme instalovat všechy soubory
# z "deployment", tentokrát do "/opt/InstallerTest"
# cílového systému (absolutní cesta)
install(DIRECTORY deployment_linux/ DESTINATION

/opt/${PROJECT_NAME})

# Navíc je třeba nainstalovat předem vytvořeného
# zástupce do "/usr/share/applications"
install(FILES linux/${PROJECT_NAME}.desktop

DESTINATION share/applications)

# a případně spustit skript po dokončení instalace
# pro nastavení práv souborů atd.
set(CPACK_DEBIAN_PACKAGE_CONTROL_EXTRA

"${CMAKE_CURRENT_SOURCE_DIR}/linux/postinst;")

Nástroj CPack – Linux (RPM)

• Tvorba RPM balíčku pomocí CPack
set(CPACK_GENERATOR RPM) # Generátor RPM

install(DIRECTORY deployment_linux/ DESTINATION
/opt/${PROJECT_NAME})

install(FILES linux/${PROJECT_NAME}.desktop
DESTINATION /usr/share/applications)

# Zde je třeba informovat, že složky "opt/"
# a "/usr/share/applications" jsou již součástí OS.
set(CPACK_RPM_EXCLUDE_FROM_AUTO_FILELIST_ADDITION

/opt /usr/share/applications)

# Opět lze specifikovat skripty prováděné
# v různých fázích instalace
set(CPACK_RPM_POST_INSTALL_SCRIPT_FILE

"${CMAKE_CURRENT_SOURCE_DIR}/linux/postinst")
set(CPACK_RPM_PRE_UNINSTALL_SCRIPT_FILE

"${CMAKE_CURRENT_SOURCE_DIR}/linux/preun")

Nástroj CPack – Linux (DEB/RPM)

• Generování balíčku

> cmake -G "Unix␣Makefiles" .
> cmake --build . --target package

• V případě RPM předpokládá nainstalovaný nástroj rpmbuild

Reference

• CMake
• https://cmake.org/cmake-tutorial/
• https://cmake.org/cmake/help/latest/

• NSIS
• https://nsis.sourceforge.io/Download
• https://nsis.sourceforge.io/Docs

